



DECUS

PROGRAM LIBRARY

DECUS NO.

8-129

TITLE

Magnetic Tape Program Library System

AUTHOR

Donald C. Uber

COMPANY

University of California, Livermore, California

DATE

July 21, 1968

SOURCE LANGUAGE

DECLASSIFIED

EXEMPT FROM AUTOMATIC DOWNGRADING AND DECLASSIFICATION



A PROGRAM LIBRARY SYSTEM FOR THE PDP-8 COMPUTER USING IBM-COMPATIBLE MAGNETIC TAPE

Abstract

This system enables the user to create a large library of programs for the PDP-8 computer which can be written on and called off IBM tape by name from the teletype. It also includes high-speed and low-speed BIN and RIM paper tape loaders, which are also called in from the teletype. The entire system uses only the last page of memory.

Programs may be subdivided into several records on tape, each with the

same name. When a program is called, all records with that label are loaded into core. This makes it possible to modify, correct, or add to programs by adding records of the same name to the library.

A switch option permits programs to be relocated in core as they are loaded in. If a program is stored as several records, they may be individually relocated.

Introduction

The PDP-8 is enjoying widespread use in the small general-purpose computer field. However, the basic machine is limited to paper tape as a means of program storage, so that the user is constantly involved with the well-known problems of low speed, mechanical reader failure, and wear and tear on his frequently used tapes. Also, unless fan-fold is used, the paper tapes are difficult to handle and store.

Those users having DEctape have a convenient means of program storage and retrieval, but no such capability exists

for PDP-8 installations with the 57A Magnetic Tape Control and IBM-compatible transports. The library system described in this report was developed to fill this need.

The library has several important features:

1. Since almost all operations are controlled from the keyboard, the system is convenient to use and is easily learned, even by those unfamiliar with the PDP-8.

2. Programs are called in by name from the teletype. This label may contain up to twelve characters. If a typing error is made, the RUBOUT key is used to delete the erroneous label.
3. Paper-tape capabilities are fully retained. BIN or RIM loaders are called off the library tape by typing the letters B or R.
4. All system operations use only the last page of memory. This is accomplished by swapping the various loaders in and out of core from the library tape.
5. Library programs may be of any length and may occupy any portion of memory except the last page.
6. When a program is called in, a switch option may be set if relocation is desired. The computer halts before loading the program in, allowing the operator to set the switch register to the new page at which he wishes to start loading the program.
7. After reading a program or loader off the library tape, the computer checks the parity error flag and halts on an error. The operator has the option of repeating the operation or ignoring the error and attempting to proceed.
8. After writing a program on the library tape, the computer reads it back and compares it word for word with the program in memory. If a discrepancy is found, the computer erases three inches of tape, and writes and checks the program again.
9. More than one program with the same name may be added to the library. When that name is called, all programs with that label are loaded in. This is useful if the program is subdivided into non-adjacent core areas. It also provides the capability of making future corrections and additions to library programs. It is possible to individually relocate the various sections of divided programs.
10. If the desired program is found and loaded in without error, the teletype prints "OK." There is no printout if no program exists with the given label.
11. Unless the relocation option is requested, tape motion is continuous during the search and load operation. This speeds up the search and reduces wear on the transport.
12. A tape with one hundred typical programs may be searched and a program loaded in about 30 sec, using a 112.5 in./sec transport. The library can hold 1365 programs, which requires a maximum of 1800 feet of tape at 800 cpi density.
13. Library material need not be executable programs. For example, the system may be used to fill memory (except the last page) with HLT, NOP, or zeros, or for data storage.

System Requirements

The library system operates with a basic PDP-8 with 4096-word memory, ASR-33 teletype, 57A Magnetic Tape Control, and a single tape transport. The system is in use on an IBM 729-VI transport. Modifications to the rewind portion

of the Bootstrap program may be required if other transports are used; however this is not certain. The selected transport is Unit 1. The programs write tape at 800 cpi with odd parity.

Library Tape Organization

Beginning at the load point, the library tape is laid out as follows:

<u>Record</u>	<u>Length (Octal PDP-8 words)</u>	
Command Record	137	} Executive Records
BIN Record	137	
Fetch Record	137	
Write Record	137	
RIM Record	137	
Label Record	14	} First Program
Location Record	14	
Program Record	1-7600	
etc.		
Label Record	14	} Last Program
Location Record	14	
Program Record	1-7600	
End of File		

The five Executive Records occupy the beginning of the tape and contain the actual routines for the library system operations. They are all the same length and are swapped in and out of the last page of core.

Programs are stored on the remainder of the tape with three records each. The

first, or Label Record, contains the twelve-character name of the program. The second, or Location Record, specifies where in memory the program is to be loaded in. The third, or Program Record, is the program itself. An End of File follows the last program.

Memory Map

The library routines occupy the last page of core as follows:

7600 - 7623	Mag Tape Bootstrap	7763	FLAG
7624 - 7762	One of the five Executive Records	7764 - 7777	Teletype label buffer

Program Operation

BOOTSTRAP

The Bootstrap resides permanently in core. When started at location 7600, it rewinds the tape, reads the Command Record into locations 7624 - 7762, and transfers control to it. When the tape starts rewinding, the teletype bell is rung. This provides a 100-msec programmed delay which is necessary before the Unit Ready Flag can be sampled (at least if the IBM 729-VI transport is used) and alerts the operator that the Command Record is about to be read in.

COMMAND RECORD

The Command Record supervises the library system. It accepts operator commands and program labels from the teletype and passes control to the other four Executive Records. When entered from the Bootstrap, the Command Record first checks the contents of location 7763 (FLAG) and prints "OK" if FLAG is non-zero. (FLAG is cleared by the Bootstrap but is later set to indicate proper execution of the Fetch, Write, and BIN operations.) The system is now awaiting a teletype command and is said to be in the Command Mode. The command may be an F, W, B, or R, depending on whether

the operator wants to Fetch or Write a program on the library tape, or use the BIN or RIM loaders. Any other letter is illegal and causes a question mark to be printed.

If commands B or R are given, the Command Record immediately loads the BIN or RIM Record in over itself. To read the BIN Record, the program simply jumps to the middle of the Bootstrap (location 7612), which reads the next record into locations 7624 - 7762 and begins executing it. The RIM Record is loaded the same way, except that the Command Record spaces forward over three records before jumping to 7612. The BIN and RIM Records are described in detail below.

When the command is F or W, the Command Record first accepts a label typed by the operator before transferring control to the Fetch or Write Record. For programming convenience, 215(8) is subtracted from label characters before they are stored in the buffer (locations 7764 - 7777). The label is terminated when the operator types either the twelfth character or the RETURN key. In the latter case, the program fills the remainder of the buffer with zeros. At this point, the program spaces the tape forward one or

two records (depending on the command) and jumps to location 7612 in the Bootstrap, which reads either the Fetch or Write Record in over the Command Record and begins executing it. The Fetch and Write Records are described in detail below.

BIN RECORD

The BIN Record operates identically to DIG-8-2-U revised 3/23/66. It will load programs into Extended Memory and provides a switch option for high-speed or low-speed readers. When it is entered via the Command Record, the BIN Record first halts to allow the operator to load his paper tape, set bit 0 of the Switch Register for high or low speed reader, and turn the reader on. When the CONTINUE key is pressed, the program reads in the BIN tape, halting at the end so the operator can turn the reader off. The AC (Accumulator) holds the checksum. When CONTINUE is pressed again, the AC is set if the checksum is zero, but cleared otherwise, and the program jumps out to the Bootstrap at 7600. The Bootstrap stores the AC in FLAG, rewinds the tape, and enters the Command Record. "OK" is now printed if FLAG is set (zero checksum), and the next command is awaited.

RIM RECORD

The RIM Record contains high-speed and low-speed RIM loaders, which the operator selects with bit 0 of the Switch Register. When entered via the Command Record, the RIM Record first halts to allow paper tape loading and switch setting.

The selected reader loads the RIM tape when CONTINUE is pressed. There is no exit from this program and the Bootstrap must be restarted at 7600 to reenter the Command Mode.

WRITE RECORD

After the operator has typed a W and the name of the program he wishes to add to the library, the Command Record replaces itself with the Write Record and begins executing it. This is the routine that writes on the tape the three records associated with that program. It begins with two halts which allow the operator to enter the IA (Initial Address) and FA (Final Address) of the program into the Switch Register. "IA/" and "FA/" are printed out to tell the operator which number to load next.

The tape is then spaced out to the EOF (End of File) and backspaced over the EOF, and the label buffer is written as the 12-word Label Record. The program then replaces the first character of the label buffer with IA-1 and writes the label buffer on tape again, this time as the 12-word Location Record. Next, the Program Record itself is written, the Word Count being computed from the IA and FA provided by the operator. After each of the three records is written, it is backspaced over and tested with a Read Compare operation. If an error is found, the faulty record is backspaced over, three inches of blank tape are written (write EOF and backspace), and the record is written and checked again. This repeats until a good record is written. Finally, new EOF is written and the program jumps out to the Bootstrap, which

rewinds the tape and loads in the Command Record. FLAG has been set by the temporary storage of -FA, which will always be non-zero (writing the last page of core on tape is forbidden), so the teletype prints "OK". The system is again in the Command Mode.

A maximum of 1365 programs can be written. This figure is dictated by the fact that when the tape is commanded to space to EOF (Space Forward Continuous with Word Count = 0), it will stop after 4096 records if EOF is not sensed first. When 1365 programs are on the tape and space to EOF is commanded, the tape will space forward 4096 records (1365×3 records per program plus one for the RIM Record) and halt just before the EOF. For this reason, the 1366th program will be written in the wrong place. However, in most applications this limit will never be approached. Even if it were, a 2400-ft reel of tape would not be exhausted, because the maximum program (locations 0000 - 7600) occupies only about 15 in. of tape, including the Label and Location Records.

FETCH RECORD

The Command Record calls and begins executing the Fetch Record when the operator has typed an F and the name of the program he wishes to load. This routine searches the tape and loads in all programs having that name. It also controls relocation of the program if the operator has requested it by setting bit 11 of the Switch Register before the tape search starts.

The Fetch Record begins in the Read Compare Continuous mode to search the

tape for the label stored in the teletype buffer. Every record is searched, beginning with the Write and RIM Records and continuing into the Label, Location, and Program Records of the library. The Word Count is cleared before each record, so that the entire record is compared against the same number of core words, starting at the label buffer. This makes it extremely unlikely that a match will be found in any but the desired label record, and in fact no such problem has occurred in our library.

At the end of each record, the Read Compare Error and End of File status bits are tested to see if the label has been found or the end of the library reached. When the label is found, the first word (IA-1) of the following record (the Location Record) is read in and temporarily stored in FLAG. Bit 11 of the Switch Register is then tested to see if the operator has requested relocation, and if so, the teletype prints "PG/" and the tape and computer halt to allow him to set the relocation page in the Switch Register. At this point the Accumulator displays the unrelocated IA. Relocation is accomplished by moving this IA to the same location on another page; it will work only for single-page programs, or multi-page programs whose indirect linkages are all on page zero. When the CONTINUE key is pressed, the relocated IA is computed by combining bits 0-4 of the Switch Register with bits 5-11 of the AC, and the Program Record is loaded in, beginning at that address.

If relocation was not requested, the Program Record is immediately loaded in at the IA specified in the Location Record. Whether the program is relocated or not, the Word Count is cleared before

the program is read in, so words continue to enter core until the end of the record is reached. Tape parity is then checked. The computer clears FLAG and halts with the Link=1 on an error. Otherwise FLAG is set and the Read Compare Continuous mode is resumed to search for other records with the same label.

Each time the label is found, the program is read in and relocated as above, so that if a program has been subdivided into several records, they may be individually relocated. This would be useful, for example, in relocating the main body of a program while leaving linkages, constants, and subroutines on page zero.

The search continues until the EOF is found. The Bootstrap is then reentered to rewind the tape and read in the Command Record. "OK" will be printed if FLAG is set, indicating that the program has been found and that the last Program Record was loaded in with no parity error. The system is again in the Command Mode. The program may then be started from the console switches in the normal manner.

Deletions from the library are impossible. However, if the user desires to correct, modify, or add to a program,

he simply adds to the library, using the same name as the original program. When the program is fetched, the new version will load in over the old one.

GENERATE PROGRAM

The Generate Program is not part of the operating library system, but is used to create new library tapes by writing the five Executive Records on them. The Generate Program and the Executive Records are first placed in core by loading the Master Generator paper tape. The Executive Records then occupy the five core pages starting at 6400, 6600, 7000, 7200, and 7400, while the Generate Program itself occupies the page starting at 6200. When started at 6200, the Generate Program writes each Executive Record on the new tape, starting at the load point, and checks each one with a Read Compare. It halts on an error with the address +1 of the error word in the Accumulator. If CONTINUE is pressed, 3 in. of blank tape are written, and the record is written and checked again. When the last record has been successfully written, the computer writes an EOF and halts. The Bootstrap can then be toggled in and programs written on the new library tape.

Operating Instructions

GENERATING A NEW LIBRARY TAPE

1. Load the Master Generator binary paper tape.
2. Hang a blank magnetic tape on the transport, with the write lock ring in. Select Unit 1.
3. LOAD ADDRESS 6200, START. The five Executive Records are written on the tape, followed by an EOF. The computer halts with the AC and Link=0. The library is ready for use.
4. If the computer halts with the AC not zero, a Read Compare error has been

found. The AC displays the address +1 of the first word in error. Press the CONTINUE key to write 3 in. of blank tape and repeat the faulty record.

USING THE LIBRARY

- 1.0 Hang the library tape on the transport and select Unit 1. Unless programs are to be written, the write lock ring should be removed.
- 2.0 Toggle the Mag Tape Bootstrap into locations 7600 - 7623.
- 3.0 LOAD ADDRESS 7600, START. The bell rings and the system enters the Command Mode.
- 4.0 Type an F to Fetch a library program, a W to Write a new program on the library, a B to use the BIN loader, or an R to use the RIM loader. Any other command is illegal and causes the computer to print a question mark and reenter the Command Mode.
- 5.0 To Fetch a program (command F was typed):
 - 5.1 Type in the exact label of the program. If an error is made, type RUBOUT and start the label over.
 - 5.2 After the last character, type RETURN to start the tape search. (If the label has 12 characters, the last one automatically initiates the search.) When the end of the library is reached, the bell rings, the tape rewinds, and the Command Mode is reentered. "OK" is printed out if the desired program was loaded in. If no program with that name was found, there is no printout. Go to step 4.0 or start the loaded program

with the console switches in the normal manner.

- 5.3 If relocation is desired, set SR (Switch Register) bit 11 to 1 before the tape search is started. Whenever the label is found, the computer prints "PG/" and halts with the unrelocated IA (Initial Address) in the AC. Set SR bits 0-4 to the relocated page address and hit the CONTINUE key. SR bits 0-4 and AC bits 5-11 form the relocated IA (SR bits 5-11 are ignored). The program is loaded in at that address and the search is resumed. Leave SR bit 11=1 to relocate subsequent records with the same label, otherwise set SR bit 11=0. Caution: Do not relocate programs into the last page of core.
- 6.0 To Write a program on the library (command W was typed):
 - 6.1 Type in any label up to twelve characters long. In case of a mistake, type RUBOUT and start a new label. After the last character, type RETURN. (If 12 characters are typed, the last automatically terminates the label.)
 - 6.2 The computer prints "IA/" and halts with the AC=0. Set the Switch Register to the Initial Address of the program and hit the CONTINUE key.
 - 6.3 The computer prints "FA/" and halts with the AC displaying the IA just loaded. Set the Switch Register to the Final Address of the program and hit the CONTINUE key. Any area of core except the last page may be

written on tape. The computer prints "OK" and reenters the Command Mode when it finishes adding the new program to the library.

- 7.0 To use the BIN Loader (command B was typed):
 - 7.1 The computer halts with the Link and AC=0. Set SR bit 0 to 1 for the low speed reader or to 0 for the high speed reader.
 - 7.2 Place the paper tape in the reader and turn the reader on (if ASR 33).
 - 7.3 Hit the CONTINUE key. The BIN tape is loaded and the computer halts with the checksum in the AC. Turn the reader off (if ASR 33).
 - 7.4 Press the CONTINUE key to reenter the Command Mode. The computer prints "OK" if the checksum was zero, nothing otherwise. If CONTINUE is not hit at this time, the BIN Record remains in memory and may be manually started at location 7627 at any time.

- 8.0 To use the RIM Loader (command R was typed):

RIM tapes are loaded the same as BIN tapes as described in steps 7.1 - 7.3. There is no checksum halt, however, and the Bootstrap must be restarted at 7600 to reenter the Command Mode. If left in core, the RIM Loader may be manually started at 7627 at any time.

9.0 Error Conditions and Recoveries

- 9.1 In general, if the computer halts in the library system with the

Link=1, an error has been detected. The one exception is the checksum halt in the BIN Loader, where the link may =1. If the Link=0 at any halt, it is not an error halt.

- 9.2 Whenever one of the five Executive Records is read into the last page of memory, the computer checks the tape parity error flag and halts on an error with the Link=1 and the AC=0. For the BIN and RIM Records, this halt is at location 7626. The computer always halts at this address so that the operator can load his paper tape, but the Link=1 only on a parity error. For the Command, Fetch, and Write Records, the error halt is at location 7627. In all cases, the error may be ignored by hitting the CONTINUE key, but it is safer to restart the Bootstrap at 7600 and try again.
- 9.3 When fetching a program from the library, the computer will halt at location 7627 with the Link=1 and the AC=0 if a parity error is found in a Program Record. The operator may ignore the error by hitting CONTINUE, or may restart the Bootstrap at 7600 and repeat the operation. Although this halt is at the same address as the parity error halt for the Fetch Record, the Fetch Record halt occurs immediately after the label is typed in, while the Program Record halt occurs after the computer has searched the tape and found the label.

Program Listing

Listings for the Bootstrap, Generate Program, and the five Executive Records are given.

The Executive Records are listed at their core addresses on the Master Generator paper tape. These are the locations they occupy when the Generator

Program writes them on a new library tape. However, in actual use the Executive Records alternately occupy the last page of core.

Each Executive Record was written and assembled as an independent one-page program. Therefore, many duplicate tags exist among the various programs.

```

/PDP-8/57A MAG TAPE LIBRARY      7/21/67 BOOTSTRAP
/SA=7600
*7600
FLAG=7763
7600 3363 BOOT, DCA FLAG /STORE FLAG (CLEAR FLAG ON 1ST PASS)
7601 1220 TAD RWND
7602 6706 BELL, MTS
7603 6716 MTC /REWIND TAPE
7604 1202 TAD BELL
7605 6046 TLS /RING BELL
7606 6041 TSF
7607 5206 JMP .-1 /WAIT 100 msec.
7610 6711 MSUR
7611 5210 JMP .-1 /WAIT TILL REWIND FINISHED
7612 7623 NEXT, 7623 /ACTS AS CLA
7613 1212 TAD NEXT
7614 6745 MCA /LOAD IA-1 INTO CA AND CLEAR WC
7615 1221 TAD RD
7616 6706 MTS /SELECT UNIT 1, ODD PARITY, 800 BPI, LOW SLICE
7617 6716 MTC /READ RECORD INTO LOCATIONS 7624-7762
7620 1101 RWND, 1101 /DUMMY COMMAND
7621 7675 RD, 7675 /ACTS AS CLA
7622 6701 MSCR
7623 5222 JMP .-1 /WAIT UNTIL RECORD LOADS IN, THEN EXECUTE.

```

```

BELL 7602
BOOT 7600
FLAG 7763
NEXT 7612
RD 7621
RWND 7620

```


*6424

NEXT=6412

6424	6734	MTRS	/ENTER FROM BOOTSTRAP
6425	7006	RTL	
6426	7630	SZL CLA	
6427	7402	HLT	/PARITY ERROR. LINK IS 1, AC IS 0.
6430	1363	START, TAD FLAG	
6431	7640	SZA CLA	
6432	5277	Q, JMP TYPK	/FLAG IS 1. TYPE "OK".
6433	4255	JMS CRLF	
6434	4255	JMS CRLF	
6435	4263	JMS READ	/COMMAND MODE. GET COMMAND CHARACTER.
6436	1311	TAD M302	
6437	7450	SNA	/B
6440	5354	JMP B	
6441	1312	TAD M4	
6442	7450	SNA	
6443	5321	JMP FETCH	/F
6444	1313	K, TAD M14	
6445	7450	SNA	
6446	5305	JMP RIM	/R
6447	1314	TAD M5	
6450	7650	SNA CLA	
6451	5320	JMP WRITE	/W
6452	1232	TAD Q	
6453	4271	JMS PRINT	/ILLEGAL CHARACTER. PRINT "?".
6454	5230	JMP START	/GET NEW COMMAND.
6455	0000	CRLF, 0	/PRINTS CR, LF
6456	1317	0, TAD CR	
6457	4271	JMS PRINT	
6460	1355	TAD LF	
6461	4271	JMS PRINT	
6462	5655	JMP I CRLF	
6463	0000	READ, 0	/GETS KEYBOARD CHARACTER
6464	6031	KSF	
6465	5264	JMP --1	
6466	6036	KRB	
6467	6046	TLS	
6470	5663	JMP I READ	
6471	0000	PRINT, 0	/PRINTS CHARACTER
6472	6041	TSF	
6473	5272	JMP --1	
6474	6046	TLS	
6475	7200	CLA	
6476	5671	JMP I PRINT	
6477	1256	TYPK, TAD 0	/PRINTS "OK"
6500	4271	JMS PRINT	
6501	1244	TAD K	
6502	4271	JMS PRINT	
6503	3363	DCA FLAG	/CLEAR FLAG
6504	5230	JMP START	/ENTER COMMAND MODE
6505	1316	RIM, TAD M3	
6506	3356	DCA FWD	/LOAD RECORD COUNTER WITH -3
6507	5342	JMP SPACE	
6510	1661	T1661, 1661	
6511	7476	M302, -302	
6512	7774	M4, -4	
6513	7764	M14, -14	
6514	7773	M5, -5	

/COMMAND RECORD

6515	7777	M1, -1	
6516	7775	M3, -3	
6517	0215	CR, 215	
6520	1315	WRITE, TAD M1	
6521	1315	FETCH, TAD M1	
6522	3356	DCA FWD	/LOAD RECORD COUNTER WITH -1 OR -2
6523	4255	LABEL, JMS CRLF	/ROUTINE ENTERS LABEL FOR FETCH OR WRITE
6524	1362	TAD LBLIA	
6525	3357	DCA ADR	/INITIALIZE LABEL ADDRESS
6526	4263	CHAR, JMS READ	/GET LABEL CHARACTER
6527	1360	TAD M377	
6530	7450	SNA	
6531	5323	JMP LABEL	/RUBOUT. RE-INITIALIZE LABEL.
6532	1361	TAD P162	
6533	7440	SZA	
6534	5337	X, JMP STORE	
6535	1334	TAD X	/CHAR WAS CR. CHANGE "JMP CHAR" TO "JMP STORES"...
6536	3341	DCA STORE+2	/... TO FILL REST OF LABEL WITH ZEROS.
6537	3757	STORE, DCA I ADR	/STORE (CHAR-215), OR ZERO IF CR WAS HIT.
6540	2357	ISZ ADR	
6541	5326	JMP CHAR	/GET NEXT CHAR, OR ANOTHER ZERO IF CR WAS HIT.
6542	6745	SPACE, MCA	/LABEL BUFFER FULL. CLEAR CA AND WC.
6543	1356	TAD FWD	
6544	6742	MRWC	/LOAD WC WITH RECORD COUNTER
6545	7200	CLA	
6546	1310	TAD T1661	
6547	6706	MTS	
6550	6716	MTC	/SPACE FORWARD 1, 2, OR 3 RECORDS
6551	6701	MSCR	
6552	5351	JMP --1	
6553	7200	CLA	
6554	4255	B, JMS CRLF	/PRINT CR, LF
6555	5212	LF, JMP NEXT	/ENTER BOOTSTRAP TO READ RECORD IN OVER THIS ONE
6556	0000	FWD, 0	
6557	0000	ADR, 0	
6560	7401	M377, -377	
6561	0162	P162, 162	
6562	7764	LBLIA, 7764	
6563	0000	FLAG, 0	

ADR	6557
B	6554
CHAR	6526
CR	6517
CRLF	6455
FETCH	6521
FLAG	6563
FWD	6556
K	6444
LABEL	6523
LBLIA	6562
LF	6555
M1	6515
M14	6513
M3	6516
M302	6511
M377	6560

/COMMAND RECORD

M4	6512
M5	6514
NEXT	6412
O	6456
PRINT	6471
P162	6561
Q	6432
READ	6463
RIM	6505
SPACE	6542
START	6430
STORE	6537
TYPOK	6477
T1661	6510
WRITE	6520
X	6534

*6624

BOOT=6600

6624	6734	MTRS	/ENTRY FROM COMMAND RECORD
6625	7006	RTL	
6626	7602	CLA HLT	/L IS 1 IF PARITY ERROR. SET SR0 FOR HI/LO READER
6627	5307	JMP BEGIN	/CONTINUE KEY STARTS BIN LOADER
6630	0000	BEGG, 0	/REMAINDER SAME AS DIG-8-2-U, REV 3/23/66,...
6631	3365	DCA SWITCH	/... EXCEPT AS NOTED.
6632	4262	JMS READ	
6633	1301	TAD M376	
6634	7750	SPA SNA CLA	
6635	5241	JMP .+4	
6636	2365	ISZ SWITCH	
6637	7040	CMA	
6640	5231	JMP BEGG+1	
6641	1365	TAD SWITCH	
6642	7640	SZA CLA	
6643	5232	JMP BEGG+2	
6644	1367	TAD CHAR	
6645	0362	AND MASK	
6646	1347	TAD M200	
6647	7510	SPA	
6650	2230	ISZ BEGG	
6651	7750	SPA SNA CLA	
6652	5630	JMP I BEGG	
6653	1367	TAD CHAR	
6654	0260	AND FMASK	
6655	1261	TAD CHANGE	
6656	3366	DCA MEMTEM	
6657	5232	JMP BEGG+2	
6660	0070	FMASK, 70	
6661	6201	CHANGE, CDF	
6662	0000	READ, 0	
6663	0000	0	
6664	6031	LOR, KSF	
6665	5264	JMP .-1	
6666	6036	KRB	
6667	3367	DCA CHAR	
6670	1367	TAD CHAR	
6671	5662	JMP I READ	
6672	6011	HIR, RSF	
6673	5272	JMP .-1	
6674	6016	RRB RFC	
6675	5267	JMP LOR+3	
6676	4351	BEND, JMS ASSEMB	
6677	7041	CIA	
6700	1370	TAD CHKSUM	
6701	7402	M376, HLT	/AC HOLDS CHECKSUM. L MAY BE 1. HIT CONT. KEY...
6702	7640	SZA CLA	/... TO RETURN TO COMMAND MODE.
6703	3305	DCA .+2	/CHECKSUM NOT 0. CHANGE "STA" TO "AND 0"
6704	6032	KCC	/CLEAR KEYBOARD FLAG
6705	7240	STA	/SET FLAG IF CHECKSUM IS 0. OTHERWISE CLEAR FLAG.
6706	5200	JMP BOOT	/BOOTSTRAP STORES FLAG AND ENTERS COMMAND MODE.
6707	6032	BEGIN, KCC	
6710	6014	RFC	
6711	6224	RIF	
6712	1261	TAD CHANGE	
6713	3366	DCA MEMTEM	
6714	7604	CLA OSR	
6715	7700	SMA CLA	

/BIN RECORD

6716	1361	TAD HIRI
6717	1360	TAD LORI
6720	3263	DCA READ+1
6721	4230	JMS BEGG
6722	5321	JMP .-1
6723	3370	GO, DCA CHKSUM
6724	1366	TAD MEMTEM
6725	3344	DCA MEMFLD
6726	1367	TAD CHAR
6727	3363	DCA WORD1
6730	4262	JMS READ
6731	3364	DCA WORD2
6732	4230	JMS BEGG
6733	5276	JMP BEND
6734	4351	JMS ASSEMB
6735	7420	SNL
6736	5344	JMP MEMFLD
6737	3371	DCA ORIGIN
6740	1363	CHEX, TAD WORD1
6741	1364	TAD WORD2
6742	1370	TAD CHKSUM
6743	5323	JMP GO
6744	0000	MEMFLD, 0
6745	3771	DCA I ORIGIN
6746	2371	ISZ ORIGIN
6747	7600	M200, 7600
6750	5340	JMP CHEX
6751	0000	ASSEMB, 0
6752	1363	TAD WORD1
6753	7106	CLL RTL
6754	7006	RTL
6755	7006	RTL
6756	1364	TAD WORD2
6757	5751	JMP I ASSEMB
6760	5264	LORI, JMP LOR
6761	0006	HIRI, HIR-LOR
6762	0300	MASK, 300
6763	0000	WORD1, 0
6764	0000	WORD2, 0
6765	0000	SWITCH, 0
6766	0000	MEMTEM, 0
6767	0000	CHAR, 0
6770	0000	CHKSUM, 0
6771	0000	ORIGIN, 0

ASSEMB	6751
BEGG	6630
BEGIN	6707
BEND	6676
BOOT	6600
CHANGE	6661
CHAR	6767
CHEX	6740
CHKSUM	6770
FMASK	6660
GO	6723
HIR	6672
HIRI	6761

/BIN RECORD

LOR	6664
LORI	6760
MASK	6762
MEMFLD	6744
MEMTEM	6766
M200	6747
M376	6701
ORIGIN	6771
READ	6662
SWITCH	6765
WORD1	6763
WORD2	6764

*7024

ROOT=7000

FLAG=7163

7024	6734	MTRS	/ENTRY FROM COMMAND RECORD. LABEL IS IN BUFFER.
7025	7006	RTL	
7026	7630	SZL CLA	
7027	7402	PARERR, HLT	/PARITY ERROR. L IS 1, AC IS 0.
7030	1256	TAD BUFIA	
7031	6745	MCA	/INITIALIZE CA AND CLEAR WC
7032	1317	TAD 13461	
7033	6706	MTS	/SELECT UNIT 1, ODD PARITY, 800 BPI, LOW SLICE
7034	6716	MTC	/CONTINUOUS READ COMPARE FOR LABEL
7035	4301	SEARCH, JMS WAIT	/WAIT FOR EOR, THEN GET TAPE STATUS
7036	7004	RAL	
7037	7710	SPA CLA	
7040	5313	JMP OUT	/EOF
7041	7420	SNL	
7042	5260	JMP FOUND	/LABEL FOUND
7043	1257	RESUME, TAD BUFIB	
7044	6745	MCA	/RE-INITIALIZE CA AND CLEAR WC
7045	7200	CLA	
7046	6724	MRD	/CHANGE TO CONTINUOUS READ COMPARE. SET DATA RQ.
7047	5235	JMP SEARCH	/RESUME SEARCH
7050	0000	PRINT, 0	/PRINTS CHARACTER
7051	6041	TSF	
7052	5251	JMP .-1	
7053	6046	TLS	
7054	7200	CLA	
7055	5650	JMP I PRINT	
7056	7763	BUFIA, 7763	
7057	7764	RUFIB, 7764	
7060	1277	FOUND, TAD LCNIA	/PROGRAM FOUND. PREPARE TO READ IN LOCATION RECORD
7061	6745	MCA	
7062	1300	TAD LCNWC	
7063	6742	MRWC	/1 WORD
7064	6724	MRC	/CHANGE TO CONTINUOUS READ. CLEAR DATA RQ.
7065	4301	JMS WAIT	/READ PROGRAM IA-1 INTO FLAG, THEN WAIT FOR EOR.
7066	7604	LAS	
7067	7010	RAR	
7070	7630	SZL CLA	/CHECK SR11
7071	5323	JMP RELOC	/RELOCATION DESIRED
7072	1363	TAD FLAG	/RELOCATION NOT DESIRED
7073	6745	MCA	/LOAD PROGRAM IA-1 INTO CA AND CLEAR WC
7074	7240	STA	
7075	6724	MRC	/CLEAR DATA RQ
7076	5351	JMP RDPGM	/READ PROGRAM IN
7077	7762	LCNIA, 7762	
7100	0015	LCNWC, -7763	
7101	0000	WAIT, 0	
7102	7240	STA	
7103	6722	MIWF	/CLEAR AND ENABLE WCO
7104	6732	MIEF	/ " " " ERF
7105	6731	MSEF	
7106	5305	JMP .-1	/WAIT FOR EOR
7107	7600	P7600, 7600	/CLA
7110	6734	MTRS	/GET TAPE STATUS
7111	7006	RTL	
7112	5701	JMP I WAIT	
7113	6702	OUT, MCD	/END OF LIBRARY. STOP TAPE.
7114	6701	MSCR	
7115	5314	JMP .-1	

			/FETCH RECORD
7116	5201	JMP BOOT+1	/BOOTSTRAP REWINDS TAPE AND ENTERS COMMAND MODE
7117	3476	T3461, 3461-7763	
7120	3561	T3561, 3561	
7121	0177	P177, 177	
7122	7777	M1, -1	
7123	6702	RELOC, MCD	/RELOCATION DESIRED. STOP TAPE.
7124	1346	TAD P	
7125	4250	JMS PRINT	
7126	1341	TAD G	
7127	4250	JMS PRINT	/PRINT "PG/"
7130	1243	TAD RESUME	
7131	4250	JMS PRINT	
7132	1363	TAD FLAG	/PROGRAM IA-1
7133	7001	IAC	
7134	7100	CLL	
7135	7402	HLT	/AC HOLDS PROGRAM IA, L IS 0. SET SR0-4 TO RELOC. PG.
7136	0321	AND P177	/HIT CONT. KEY TO READ IN AND RELOCATE PROGRAM
7137	3363	DCA FLAG	/STORE NON-PAGE BITS OF IA
7140	7604	LAS	
7141	0307	G, AND P7600	/AC HOLDS RELOCATION PAGE
7142	1363	TAD FLAG	/AC HOLDS RELOCATED IA
7143	1322	TAD M1	
7144	6745	MCA	/LOAD RELOCATED IA-1 INTO CA AND CLEAR WC
7145	7200	CLA	
7146	1320	P, TAD T3561	
7147	6706	MTS	/SELECT UNIT 1, ODD PARITY, 800 BPI, LOW SLICE
7150	6716	MTC	/READ IN PROGRAM IN CONTINUOUS MODE
7151	4301	RDPGM, JMS WAIT	/WAIT FOR EOR, THEN GET TAPE STATUS
7152	7620	SNL CLA	
7153	5357	JMP PAROK	
7154	6702	MCD	/PARITY ERROR. STOP TAPE.
7155	3363	DCA FLAG	/CLEAR FLAG
7156	5227	JMP PARERR	
7157	7240	PAROK, STA	/PARITY OK
7160	3363	DCA FLAG	/SET FLAG
7161	5243	JMP RESUME	/PREPARE TO RESUME SEARCH FOR LABEL

BOOT	7000
BUFIA	7056
BUFIB	7057
FLAG	7163
FOUND	7060
G	7141
LCNIA	7077
LCNWC	7100
M1	7122
OUT	7113
P	7146
PARERR	7027
PAROK	7157
PRINT	7050
P177	7121
P7600	7107
RDPGM	7151
RELOC	7123
RESUME	7043
SEARCH	7035
T3461	7117
T3561	7120
WAIT	7101

*7224

BOOT=7200

FLAG=7363

7224	6734	MTRS	/ENTRY FROM COMMAND RECORD. LABEL IS IN BUFFER.
7225	7006	RTL	
7226	7630	SZL CLA	
7227	7402	HLT	/PARITY ERROR. L IS 1, AC IS 0.
7230	1267	TAD P3	
7231	4307	JMS TYP0UT	/PRINT "IA/"
7232	7402	HLT	/SET SR TO IA, HIT CONT.
7233	7604	LAS	
7234	3270	DCA IA	/STORE IA
7235	4307	JMS TYP0UT	/PRINT "FA/"
7236	7606	LAS HLT	/AC HOLDS IA. L IS 0. SET SR TO FA, HIT CONT.
7237	7604	LAS	
7240	7041	CIA	
7241	3363	DCA FLAG	/STORE -FA. THIS ALSO SETS FLAG.
7242	1305	TAD T1661	
7243	4341	JMS TAPE	/SPACE TO EOF
7244	1306	TAD T1721	
7245	4341	JMS TAPE	/BACKSPACE OVER EOF
7246	1271	TAD BUFIA	
7247	3357	DCA TIA	/INITIALIZE TAPE IA FOR LABEL BUFFER
7250	7001	IAC	
7251	3360	DCA TWC	/ " " WC " "
7252	4317	JMS WTCHK	/WRITE AND CHECK LABEL RECORD
7253	1270	TAD IA	
7254	1272	TAD M1	
7255	3364	DCA FLAG+1	/REPLACE 1ST LABEL CHAR WITH PROGRAM IA-1
7256	4317	JMS WTCHK	/WRITE AND CHECK LOCATION RECORD
7257	1364	TAD FLAG+1	
7260	3357	DCA TIA	/PROGRAM IA-1
7261	1363	TAD FLAG	
7262	3360	DCA TWC	/-FA
7263	4317	JMS WTCHK	/WRITE AND CHECK PROGRAM RECORD
7264	1303	TAD T1321	
7265	4341	JMS TAPE	/WRITE EOF
7266	5201	JMP BOOT+1	/BOOTSTRAP REWINDS TAPE AND ENTERS COMMAND MODE
7267	0003	P3, 3	
7270	0000	IA, 0	
7271	7763	BUFIA, 7763	
7272	7777	M1, -1	
7273	0000	PRINT, 0	/PRINTS CHARACTER
7274	6041	TSF	
7275	5274	JMP .-1	
7276	6046	TLS	
7277	7200	CLA	
7300	5673	JMP I PRINT	
7301	0257	SLASH, 257	
7302	1221	T1221, 1221	
7303	1321	T1321, 1321	
7304	1421	T1421, 1421	
7305	1661	T1661, 1661	
7306	1721	T1721, 1721	
7307	0000	TYP0UT, 0	/PRINTS "IA/" OR "FA/"
7310	1322	TAD F	
7311	4273	JMS PRINT	
7312	1314	TAD A	
7313	4273	JMS PRINT	

/WRITE RECORD

7314	1301	A, TAD SLASH	
7315	4273	JMS PRINT	
7316	5707	JMP I TYP0UT	
7317	0000	WTCHK, 0	
7320	1302	TAD T1221	
7321	4341	JMS TAPE	/WRITE RECORD
7322	1306	F, TAD T1721	
7323	4341	JMS TAPE	/BACKSPACE OVER RECORD
7324	1304	TAD T1421	
7325	4341	JMS TAPE	/READ COMPARE RECORD
7326	6734	MTRS	/GET TAPE STATUS
7327	7006	RTL	
7330	7700	SMA CLA	
7331	5717	JMP I WTCHK	/NO ERROR
7332	1306	TAD T1721	/READ COMPARE ERROR
7333	4341	JMS TAPE	/BACKSPACE OVER BAD RECORD
7334	1303	TAD T1321	/WRITE...
7335	4341	JMS TAPE	/... 3"
7336	1306	TAD T1721	/... BLANK
7337	4341	JMS TAPE	/... TAPE.
7340	5320	JMP WTCHK+1	/WRITE AND CHECK RECORD AGAIN
7341	0000	TAPE, 0	/BASIC TAPE SUBROUTINE
7342	3361	DCA TCM	/STORE TAPE COMMAND
7343	1357	TAD TIA	
7344	6745	MCA	/LOAD CA
7345	1360	TAD TWC	/COMPUTE: -WC = TIA / TWC
7346	6742	MRWC	/LOAD WC
7347	7200	CLA	
7350	1361	TAD TCM	
7351	6706	MTS	/SELECT UNIT 1, ODD PARITY, 800 BPI, HIGH SLICE
7352	6716	MTC	/EXECUTE TAPE FUNCTION
7353	6701	MSCR	
7354	5353	JMP .-1	/WAIT UNTIL THROUGH
7355	7200	CLA	
7356	5741	JMP I TAPE	
7357	0000	TIA, 0	
7360	0000	TWC, 0	
7361	0000	TCM, 0	

A	7314
BOOT	7200
BUFIA	7271
F	7322
FLAG	7363
IA	7270
MI	7272
PRINT	7273
P3	7267
SLASH	7301
TAPE	7341
TCM	7361
TIA	7357
TWC	7360
TYP0UT	7307
T1221	7302
T1321	7303
T1421	7304
T1661	7305
T1721	7306
WTCHK	7317

*7424
7424 6734 MTRS /ENTRY FROM COMMAND RECORD
7425 7006 RTL
7426 7602 CLA HLT /L IS 1 IF PARITY ERROR. SET SR0 FOR HI/LO READER.
7427 7604 LAS /CONT. STARTS RIM LOADER
7430 7700 SMA CLA
7431 5253 JMP HIR
7432 6032 LOR, KCC /LOW SPEED RIM LOADER
7433 6031 KSF
7434 5233 JMP .-1
7435 6036 KRB
7436 7106 CLL RTL
7437 7006 RTL
7440 7510 SPA
7441 5233 JMP LOR+1
7442 7006 RTL
7443 6031 KSF
7444 5243 JMP .-1
7445 6034 KRS
7446 7420 SNL
7447 3652 DCA I TEMP
7450 3252 DCA TEMP
7451 5232 JMP LOR
7452 0000 TEMP,0
7453 6014 HIR, RFC /HIGH SPEED RIM LOADER
7454 6011 RSF
7455 5254 JMP .-1
7456 6016 RRB RFC
7457 7106 CLL RTL
7460 7006 RTL
7461 7510 SPA
7462 5271 JMP REPEAT-1
7463 7006 RTL
7464 6011 RSF
7465 5264 JMP .-1
7466 6016 RRB RFC
7467 7420 SNL
7470 3652 DCA I TEMP
7471 3252 DCA TEMP
7472 5254 REPEAT, JMP HIR+1

HIR 7453
LOR 7432
REPEAT 7472
TEMP 7452

6200	1377	TAD (6423	
6201	4215	JMS WTCHK	/WRITE AND CHECK COMMAND RECORD
6202	1376	TAD (6623	
6203	4215	JMS WTCHK	/ " " " BIN "
6204	1375	TAD (7023	
6205	4215	JMS WTCHK	/ " " " FETCH "
6206	1374	TAD (7223	
6207	4215	JMS WTCHK	/ " " " WRITE "
6210	1373	TAD (7423	
6211	4215	JMS WTCHK	/ " " " RIM "
6212	1372	TAD (1321	
6213	4243	JMS TAPE	/WRITE EOF
6214	7402	HLT	/ALL DONE. AC IS 0.
6215	0000	WTCHK, 0	
6216	3265	DCA TIA	/STORE IA-1
6217	1371	TAD (1221	
6220	4243	JMS TAPE	/WRITE RECORD
6221	1370	TAD (1721	
6222	4243	JMS TAPE	/BACKSPACE OVER RECORD
6223	1367	TAD (1421	
6224	4243	JMS TAPE	/READ COMPARE RECORD
6225	6734	MTRS	/GET TAPE STATUS
6226	7006	RTL	
6227	7700	SMA CLA	
6230	5615	JMP I WTCHK	/NO ERROR
6231	6744	MRCA	/READ COMPARE ERROR. GET CA.
6232	7402	HLT	/AC=ADDRESS+1 OF ERROR WORD
6233	7200	CLA	/HIT CONT. TO TRY AGAIN
6234	1370	TAD (1721	
6235	4243	JMS TAPE	/BACK SPACE OVER BAD RECORD
6236	1372	TAD (1321	/WRITE,...
6237	4243	JMS TAPE	/... 3"
6240	1370	TAD (1721	/... BLANK
6241	4243	JMS TAPE	/... TAPE.
6242	5217	JMP WTCHK+2	/WRITE AND CHECK RECORD AGAIN
6243	0000	TAPE, 0	/BASIC TAPE SUBROUTINE
6244	3264	DCA TCM	/STORE COMMAND
6245	1265	TAD TIA	
6246	6745	MCA	/LOAD CA
6247	7200	CLA	
6250	1366	TAD (-137	
6251	6742	MRWC	/LOAD WC WITH -137(8)
6252	7200	CLA	
6253	1264	TAD TCM	
6254	6706	MTS	/SELECT UNIT 1, ODD PARITY, 800 BPI, LOW SLICE
6255	6711	MSUR	
6256	5255	JMP -1	
6257	6716	MTC	/EXECUTE TAPE FUNCTION
6260	6701	MSCR	
6261	5260	JMP -1	/WAIT UNTIL THROUGH
6262	7200	CLA	
6263	5643	JMP I TAPE	
6264	0000	TCM, 0	
6265	0000	TIA, 0	
6366	7641	\$	
6367	1421		
6370	1721		
6371	1221		

/GENERATE PROGRAM

6372 1321
6373 7423
6374 7223
6375 7023
6376 6623
6377 6423

TAPE 6243
TCM 6264
TIA 6265
WTCHK 6215

Distribution

LRL Internal Distribution

Michael M. May	
D. C. Uber	25
Bio-Med Library	
D. Warner	40
TID Berkeley	
TID File	30

External Distribution

* A. J. Cossette Digital Equipment Corp. Maynard, Mass.	2
** R. de John R. McInturff Digital Equipment Corp. Palo Alto, Calif.	3
TID-4500 Distribution, UC-48, Biology and Medicine	267

LEGAL NOTICE

This report was prepared as an account of Government sponsored work. Neither the United States nor the Commission, nor any person acting on behalf of the Commission:

A. Makes any warranty or representation, expressed or implied, with respect to the accuracy, completeness, or usefulness of the information contained in this report or that the use of any information, apparatus, method, or process disclosed in this report may not infringe privately owned rights; or

B. Assumes any liabilities with respect to the use of or for damages resulting from the use of any information, apparatus, method or process disclosed in this report.

As used in the above "person acting on behalf of the Commission" includes any employee or contractor of the Commission or employee of such contractor to the extent that such employee or contractor of the Commission, or employee of such contractor prepares, disseminates or provides access to, any information pursuant to his employment or contract with the Commission, or his employment with such contractor.

Printed in USA. Available from the Clearinghouse for Federal Scientific and Technical Information, National Bureau of Standards, U. S. Department of Commerce, Springfield, Virginia 22151
Price: Printed Copy \$3.00; Microfiche \$0.65.

* Computer paper tapes and a submittal form are to be included. These will be supplied by the author. This material will be published in abstract and made available by Digital Equipment to other members of DECUS (Digital Equipment Corp. User's Group).

** Paper Tapes will be included (no submittal form).

VE/lh